Étude de fonction (2)

f est la fonction définie pour tout réel x par :

$$f(x) = \frac{x \operatorname{ch} x - \operatorname{sh} x}{\operatorname{ch} x - 1} \quad \text{si} \quad x \neq 0, \quad f(0) = \ell$$

où ℓ est un réel. On note $\mathscr{C}$ la courbe représentative de f dans un repère orthonormé.

1/ Déterminer le développement limité de f à l'ordre 3 au voisinage de 0. En déduire ℓ pour que f soit continue en 0. Dans la suite, on donne à ℓ cette valeur.

2/ Montrer que f est dérivable en 0 et préciser la position de $\mathscr{C}$ par rapport à sa tangente au point d'abscisse 0.

3/ Montrer que f est croissante sur $[0, +\infty[$.

4/ Préciser la droite asymptote à $\mathscr{C}$ au voisinage de $+\infty$ et préciser la position de $\mathscr{C}$ par rapport à cette asymptote. Tracer $\mathscr{C}$.

```
> block(f(x):=(x*cosh(x)-sinh(x))/(cosh(x)-1),f(x));
```

$$\frac{x \operatorname{ch} x - \operatorname{sh} x}{\operatorname{ch} x - 1}$$

1/ Un rapide calcul nous indique que le numérateur et le dénominateur de l'expression $f(x)$ sont d'ordre 3 et 2 en x au voisinage de 0. Pour obtenir un développement limité à l'ordre 3 en 0 de $f(x)$ il faut donc anticiper la simplification par x^2 et développer le numérateur et le dénominateur à l'ordre 5. Enfin, si on devait le faire à la main…

```
> taylor(f(x),x,0,3);
```

$$\frac{2\,x}{3}+\frac{x^3}{90}+\cdots$$

La limite de $f(x)$ est donc 0, il suffit de poser $\ell = 0$ pour que f soit continue en 0.

```
> c:limit(f(x)/x,x,0);
```

$$\frac{2}{3}$$

2/ f est dérivable en 0 (ce que l'on pouvait déduire du développement précédent) et $f'(0) = \dfrac{2}{3}$.

```
> taylor(f(x)-c*x,x,0,3);
```

$$+\frac{x^3}{90}+\cdots$$

La différence $f(x) - \dfrac{2}{3}x$ est équivalente à $\dfrac{1}{90}x^3$ au voisinage de 0, la courbe représentative de f traverse donc sa tangente à l'origine, elle passe de dessous au dessus (point d'*inflexion*).

```
> fp:diff(f(x),x);
```

$$\frac{x\,\mathrm{sh}\,x}{\mathrm{ch}\,x-1}-\frac{(x\,\mathrm{ch}\,x-\mathrm{sh}\,x)\,\mathrm{sh}\,x}{(\mathrm{ch}\,x-1)^2}$$

3/ Le signe de la dérivée n'est pas simple à déterminer sous cette forme, on factorise !

```
> factor(fp);
```

$$\frac{\mathrm{sh}\,x\,(\mathrm{sh}\,x-x)}{(\mathrm{ch}\,x-1)^2}$$

Là, les choses sont plus nettes. La quantité dont le signe n'est pas *immédiat* est $\mathrm{sh}(x)-x$, on a toutefois vite fait de se convaincre qu'elle est positive sur $\mathbf{R}_+$, en s'appuyant sur le signe de sa dérivée qui est manifestement positive. La fonction f est croissante sur $[0,+\infty[$.

```
> a:limit(f(x)/x,x,inf);
```

$$1$$

```
> b:limit(f(x)-a*x,x,inf);
```

$$-1$$

4/ Les deux calculs précédents prouvent l'existence d'une droite asymptote à $\mathscr{C}$, son équation est $y = x - 1$.

```
> block(g(x):=f(x)-a*x-b,g(x));
```

$$\frac{x\,\mathrm{ch}\,x - \mathrm{sh}\,x}{\mathrm{ch}\,x - 1} - x + 1$$

L'étude de la position de la courbe par rapport à son asymptote au voisinage de $+\infty$ peut être faite en recherchant un équivalent de $g(x) = f(x) - x + 1$. On commence par utiliser les écritures, à l'aide d'exponentielles, de ch et sh.

```
> exponentialize:true$
> A:factor(g(x));
```

$$\frac{2\left(x\,e^x - e^x + 1\right)}{\left(e^x - 1\right)^2}$$

La factorisation de $g(x)$ *prépare* le développement à suivre.

```
> taylor(A,x,inf,1);
```

$$+\left(2\,x - 2 + \cdots\right)\,e^{-x} + \cdots$$

Un équivalent de $g(x)$ au voisinage de $+\infty$ est $2xe^{-x}$ qui est positif. La courbe $\mathscr{C}$ est donc au dessus de son asymptote vers $+\infty$.


y
x
+1
+1
𝒞
y = x − 1
y = x + 1